# ===========================================================================
#      http://www.gnu.org/software/autoconf-archive/ax_boost_system.html
# ===========================================================================
#
# SYNOPSIS
#
#   AX_BOOST_SYSTEM
#
# DESCRIPTION
#
#   Test for System library from the Boost C++ libraries. The macro requires
#   a preceding call to AX_BOOST_BASE. Further documentation is available at
#   <http://randspringer.de/boost/index.html>.
#
#   This macro calls:
#
#     AC_SUBST(BOOST_SYSTEM_LIB)
#
#   And sets:
#
#     HAVE_BOOST_SYSTEM
#
# LICENSE
#
#   Copyright (c) 2008 Thomas Porschberg <thomas@randspringer.de>
#   Copyright (c) 2008 Michael Tindal
#   Copyright (c) 2008 Daniel Casimiro <dan.casimiro@gmail.com>
#
#   Copying and distribution of this file, with or without modification, are
#   permitted in any medium without royalty provided the copyright notice
#   and this notice are preserved. This file is offered as-is, without any
#   warranty.

#serial 17

AC_DEFUN([AX_BOOST_SYSTEM],
    [
	AC_ARG_WITH([boost-system],
	    AS_HELP_STRING([--with-boost-system@<:@=special-lib@:>@],
	        [use the System library from boost - it is possible to
		 specify a certain library for the linker
		 e.g. --with-boost-system=boost_system-gcc-mt ]),
	    [
		if test "$withval" = "no"; then
		    want_boost="no"
		elif test "$withval" = "yes"; then
		    want_boost="yes"
		    ax_boost_user_system_lib=""
		else
		    want_boost="yes"
		    ax_boost_user_system_lib="$withval"
		fi
	    ],
	    [want_boost="yes"]
	)

	if test "x$want_boost" = "xyes"; then
	    AC_REQUIRE([AC_PROG_CC])
	    AC_REQUIRE([AC_CANONICAL_BUILD])

	    CPPFLAGS_SAVED="$CPPFLAGS"
	    CPPFLAGS="$CPPFLAGS $BOOST_CPPFLAGS"
	    export CPPFLAGS

	    LDFLAGS_SAVED="$LDFLAGS"
	    LDFLAGS="$LDFLAGS $BOOST_LDFLAGS"
	    export LDFLAGS

	    AC_CACHE_CHECK(
	        for Boost::System headers,
		ax_cv_boost_system,
		[
		    AC_LANG_PUSH([C++])
		    CXXFLAGS_SAVE=$CXXFLAGS

		    AC_COMPILE_IFELSE(
		        [AC_LANG_PROGRAM(
			    [[@%:@include <boost/system/error_code.hpp>]],
			    [[boost::system::system_category]])],
			ax_cv_boost_system=yes,
			ax_cv_boost_system=no)
		    CXXFLAGS=$CXXFLAGS_SAVE
		    AC_LANG_POP([C++])
		])
	    if test "x$ax_cv_boost_system" = "xyes"; then
	        AC_SUBST(BOOST_CPPFLAGS)

		AC_DEFINE(
		    HAVE_BOOST_SYSTEM,,
		    [define if the Boost::System library is available])
		BOOSTLIBDIR=$(echo $BOOST_LDFLAGS | sed -e "s/@<:@^\/@:>@*//")

		LDFLAGS_SAVE=$LDFLAGS
		if test "x$ax_boost_user_system_lib" = "x"; then
		    ax_lib=boost_system
		    AC_CHECK_LIB(
			$ax_lib,
			exit,
			[
			    BOOST_SYSTEM_LIB="-l$ax_lib"
			    AC_SUBST(BOOST_SYSTEM_LIB)
			    link_system="yes"
			    break
			],
			[link_system="no"])

		else
		    for ax_lib in \
		            $ax_boost_user_system_lib \
		            boost_system-$ax_boost_user_system_lib; do
		        AC_CHECK_LIB(
			    $ax_lib,
			    exit,
			    [
			        BOOST_SYSTEM_LIB="-l$ax_lib"
				AC_SUBST(BOOST_SYSTEM_LIB)
				link_system="yes"
				break
			    ],
			    [link_system="no"])
		    done

		fi
		# Earlier versions of libboost-thread didn't need
		# libboost-system, so don't error out if they're not
		# available.  The right way to check this is to detect
		# the boost version and decide based on that.  Or else
		# wait until Wheezy support is dropped, and re-enable
		# this.

		# if test "x$link_system" = "xno"; then
		#     AC_MSG_ERROR(Could not link against $ax_lib !)
		# fi
	    fi

	    CPPFLAGS="$CPPFLAGS_SAVED"
	    LDFLAGS="$LDFLAGS_SAVED"
	fi
    ])
